

3. Syntax



3. Syntax

What is a Language?

Concrete Syntax

Abstract Syntax

List Structures in Syntax

What is a Language?

Language: A system of communication in a structured way

Natural language

- *used for arbitrary communication*
- *complex, nuanced, and imprecise*

*English, Chinese,
Hindi, Spanish, ...*

Programming language

- *used to describe computation*
- *programs have **structure** and **meaning***

*Elm, Java, C, Python,
SQL, XML, ...*

Object vs. Metalanguage

Important to distinguish two *kinds of languages*:

- **Object language:** the language we're defining
- **Metalanguage:** the language we're using to define the structure and meaning of the object language!

A single language can fill both roles at different times!
Examples: English, Elm



Syntax vs. Semantics & Metalanguages

Two main *aspects of a language*:

- **Syntax**: the structure of its programs
- **Semantics**: the meaning of its programs

Scope of Metalanguages				
	Syntax	Denotational Semantics	Operational Semantics	Type Systems
Regular Expressions	●			
Grammars	●			
Elm	●	●		●
Inference Rules	●	●	●	●

3. Syntax

What is a Language?

Concrete Syntax

Abstract Syntax

List Structures in Syntax

Grammar Metalanguage

Grammar Concepts

- **Grammar:** Set of **productions** (or **rules**)
- **Production:** $L ::= R$ where
 - L : nonterminal symbol (in a context-free grammar)
 - R : sequence of terminal & nonterminal symbols
- **Derivation:** Sequence of substitutions (“ L by R ”)
- **Sentence:** Sequence of terminals derivable from nonterminal
- **Language:** Set of derivable sentences

Context-Free Grammars

Formal Grammar Definition

A **grammar** is a four-tuple (N, Σ, P, S) where:

- N is a set of **nonterminal symbols**
- Σ is a set of **terminal symbols** with $N \cap \Sigma = \emptyset$
- $P \subseteq N \times (N \cup \Sigma)^*$ is a set of **productions**
- $S \in N$ is the **start symbol**.

Grammar for Binary Numbers

$(\{dig, bin\}, \{0, 1\}, \{(dig, 0), (dig, 1), (bin, dig), (bin, dig\ bin)\}, bin)$

Backus-Naur Form (BNF)

Grammar for Binary Numbers

$(\{dig, bin\}, \{0, 1\}, \{(dig, 0), (dig, 1), (bin, dig), (bin, dig\ bin)\}, bin)$

BNF \approx Only Productions

$dig ::= 0$

$dig ::= 1$

$bin ::= dig$

$bin ::= dig\ bin$

Grouping RHSs

$dig ::= 0 \mid 1$

$bin ::= dig \mid dig\ bin$

Example Derivations

Binary numbers

dig ::= 0 (P1)

dig ::= 1 (P2)

bin ::= *dig* (P3)

bin ::= *dig bin* (P4)

bin

\Rightarrow *dig bin* (P4)

\Rightarrow *dig dig bin* (P4)

\Rightarrow *dig dig dig* (P3)

\Rightarrow 1 *dig dig* (P2)

\Rightarrow 1 0 *dig* (P1)

\Rightarrow 1 0 1 (P2)

bin

\Rightarrow *dig bin* (P4)

\Rightarrow *dig dig bin* (P4)

\Rightarrow *dig dig dig* (P3)

\Rightarrow *dig dig* 1 (P2)

\Rightarrow *dig* 0 1 (P1)

\Rightarrow 1 0 1 (P2)

Note: One sentence may have different derivations!

Derivation \approx Trace
(only substitutions, no simplifications)

Grammars Define Languages

Language

Grammar $G = (N, \Sigma, P, S)$ defines the **language** $L(G) = \{w \in \Sigma^* \mid S \Rightarrow^* w\}$

Example

$L(\{dig, bin\}, \{0, 1\}, \{(dig, 0), (dig, 0), \dots\}, bin) = \{b^k \mid b \in \{0, 1\} \wedge k > 0\}$

Exercise ...



Question 1

Consider the language L defined by the following grammar.

$$S ::= A B$$
$$A ::= 0A \mid 0$$
$$B ::= 1B \mid 1$$

Which of the following statements about are true for words/sentences of L ? Each sentence contains ...

(a) ... one or more 1s

(e) ... exactly as many 1s as 0s

(b) ... one or more 0s

(f) ... at least two digits

(c) ... at least as many 1s as 0s

(g) All 0s precede all 1s

(d) ... at least as many 0s as 1s

Sentence Structure = Parse Tree


Binary numbers

$dig ::= 0 \mid 1$

$bin ::= dig \mid dig\ bin$

Rules can be interpreted as instructions to build trees:

“Add R as children to L ”



Internal nodes:
Nonterminals

Leaves:
Terminals

Parse tree ignores the order of rule applications
⇒ Appropriate representation of sentence structure


Exercise ...

Question 2

Create a parse tree for the following sentence: **while not T do {skip ; skip}**

Concrete Syntax

```
cond ::= T
      | not cond
      | ( cond )
stmt ::= skip
      | while cond do { stmt }
      | stmt ; stmt
```



Ambiguity

Ambiguous Grammar

Some sentences have more than one parse tree.

Ambiguous Grammar


$num ::= 0 \mid 1 \mid 2 \mid \dots$

$expr ::= num$

$\quad \mid expr + expr$

$\quad \mid expr * expr$

Parse tree for $2 + 3 * 4$



In a Nutshell ...

What is a Grammar?

A formalism to define linear representations
for **typed** tree data structures

*Each nonterminal (\approx type)
denotes a set of trees
with a specific structure.*

Parser: Algorithm to recover tree structures from strings

Pretty Printer: Algorithm to turn tree structures into strings

Why Grammar Matters

Video Clip

WARNING: This video contains R-rated language!

3. Syntax

What is a Language?

Concrete Syntax

Abstract Syntax

List Structures in Syntax


Programs are Trees!

Abstract Syntax Tree (AST)


The essential structure of a program
(everything needed to determine its semantics)

Internal nodes: **Operations**

Leaves: **Values**



AST for $2 + 3 * 4$



Parse tree for $2 + 3 * 4$


More AST Examples

Abstract Syntax Tree (AST)


The essential structure of a program
(everything needed to determine its semantics)

Internal nodes: **Operations**


Leaves: **Values**



`2 + 3 * 4`



`(5 + 6) * (7 + 8)`



`if true then 2+3 else 5`

Concrete vs. Abstract Syntax

	Concrete Syntax	Abstract Syntax
Program	Sequence of words	Tree of constructors & values
Language	Set of sentences	Set of ASTs
Description	Context-free grammar	<i>Elm type definitions</i>
Structure	Parse tree	AST = <i>Value of an Elm type</i>
Ambiguity	can be a problem	ruled out by formalism
Keywords	yes	no
Parentheses	yes	no ¹
Convenience		smaller in size

¹Parentheses of the metalanguage (i.e., Elm) may be used
Abstract Syntax

Abstract Syntax via Data Types

Concrete Syntax

```
term ::= true  
      | false  
      | not term  
      | if term then term else term  
      | ( term )
```

Abstract Syntax

```
type Term = Tru  
        | Fls  
        | Not Term  
        | If Term Term Term
```

Translation

Nonterminal \longrightarrow Type

Terminal \longrightarrow Value / Constructor

Production RHS \longrightarrow Constructor (with arguments)

Nonterminals on RHS \longrightarrow Type arguments for constructor

Example

term \longrightarrow **Term**

true \longrightarrow **Tru**

not *term* \longrightarrow **Not Term**

term \longrightarrow **Term**

Exercise ...

Concrete Syntax

```
num ::= 0 | 1 | 2 | ...  
expr ::= num  
        | -expr  
        | expr + expr  
        | expr * expr  
        | ( expr )
```

Question 3

Define a data type **Expr** to represent the abstract syntax.

Solution to Exercise

Concrete Syntax

```
num ::= 0 | 1 | 2 | ...  
expr ::= num  
        | -expr  
        | expr + expr  
        | expr * expr  
        | ( expr )
```

Abstract Syntax

```
type Expr = Num Int  
        | Neg Expr  
        | Plus Expr Expr  
        | Times Expr Expr
```

Note: The `Num` constructor is needed to embed `Int` into `Expr`.

Program = Tree = Data Type Value


Linear/textual form

2 + 3 * 4


Elm values

```
Plus (Num 2) (Times (Num 3) (Num 4))  
Times (Plus (Num 2) (Num 3)) (Num 4)
```

Informal AST notation



Elm ASTs




Abstract Syntax Trees vs. Parse Trees

Ambiguous Grammar

$num ::= 0 \mid 1 \mid 2 \mid \dots$

$expr ::= num \mid expr + expr \mid expr * expr$



2 parse trees for

Abstract Syntax


2 + 3 * 4

Abstract Syntax

type Expr = Num Int

| Plus Expr Expr

| Times Expr Expr



Forced Unique ASTs for

2 + 3 * 4

Grammars into Data Types

Grammar		Data Type	
Basic nonterminal	<i>num, name, ...</i>	Predefined type	<i>Int, String, ...</i>
Non-basic nonterminal	<i>expr, term, ...</i>	Data type	<i>Expr, Term, ...</i>
Terminal for operation	+ , if , ...	Constructor	<i>Plus, If, ...</i>
Grouping/filler terminal	begin , (,) , do , ...	—	—
“Subtype” nonterminal	<i>num</i>	Extra constructor	<i>Num Int</i>

Nonterminals \approx **Types**

Terminals \approx **Values / Operations**

Exercise ...

Concrete Syntax

```
cond ::= T
        | not cond
        | ( cond )
stmt ::= skip
        | while cond do { stmt }
        | stmt ; stmt
```

Question 4

Define the data types **Cond** and **Stmt** to represent the abstract syntax.

Solution to Exercise

Concrete Syntax

```
cond ::= T  
        | not cond  
        | ( cond )  
stmt ::= skip  
        | while cond do { stmt }  
        | stmt ; stmt
```

Abstract Syntax

```
type Cond = T | Not Cond  
  
type Stmt = Skip  
          | While Cond Stmt  
          | Seq Stmt Stmt
```

Exercise ...

Question 5

Translate the following program into abstract syntax, i.e., write it as a value of type `Stmt`.

```
while not(not(T)) {  
  while T {  
    skip; skip  
  }  
}
```

Abstract Syntax

type Cond = T | Not Cond

type Stmt = Skip
 | While Cond Stmt
 | Seq Stmt Stmt


```
While (Not (Not T))  
  (While T  
    (Seq Skip Skip))
```

Exercise ...

Question 6

Draw the *abstract syntax tree* for the following program.

```
while not(not(T)) {  
  while T {  
    skip; skip  
  }  
}
```




Exercise ...

Question 7

Draw the *abstract syntax tree* for the following program.

```
while not(not(T)) {  
  while T {skip}; skip  
}
```



3. Syntax

What is a Language?

Concrete Syntax

Abstract Syntax

List Structures in Syntax

Syntactic Repetition via Recursion

Concrete Syntax

```
 $n \in name ::= (strings)$   
 $fundef ::= n\ pars = expr$   
 $pars ::= n\ pars \mid \epsilon$ 
```

Abstract Syntax

```
type alias Name = String  
type FunDef = Fun Name Pars Expr  
type Pars = P Name Pars | Empty
```

*ϵ -rule allows removal of $pars$
nonterminal*

```
f m x y = m*x + y
```

```
Fun "f" (P "m" (P "x" (P "y" Empty))) (Plus (Times ...) ...)
```

Syntactic Repetition via Lists

Concrete Syntax

```
 $n \in \text{name} ::= (\text{strings})$   
 $\text{fundef} ::= n\ n^* = \text{expr}$ 
```

Abstract Syntax

```
type alias Name = String  
type FunDef = Fun Name (List Name) Expr
```

```
f m x y = m*x + y
```

```
Fun "f" ["m","x","y"] (Plus (Times ...) ...)
```

Representing Non-Zero Repetitions (A)

Concrete Syntax

```
 $n \in name ::= (strings)$   
 $fundef ::= n\ pars = expr$   
 $pars ::= n\ pars \mid n$ 
```

Abstract Syntax (A)

```
type alias Name = String  
type FunDef = Fun Name Pars Expr  
type Pars = P Name Pars | N Name
```

```
f m x y = m*x + y
```

```
Fun "f" (P "m" (P "x" (N "y"))) (Plus (Times ...) ...)
```

Representing Non-Zero Repetitions (B)

Concrete Syntax

```
 $n \in \text{name} ::= (\text{strings})$   
 $\text{fundef} ::= n \text{ pars} = \text{expr}$   
 $\text{pars} ::= n \text{ pars} \mid n$ 
```

Abstract Syntax (B)

```
type alias Name = String  
type alias FunDef = (Name, Pars, Expr)  
type alias Pars = (Name, List Name) -- precise  
type alias Pars = List Name        -- also ok
```

```
f m x y = m*x + y
```

```
("f", ["m", "x", "y"], Plus (Times ...))
```

Discussion ...



The following type definition represents the abstract syntax of a specific form of integer lists.

```
type S = A Int  
        | B Int Int S
```

Which type corresponds most closely to S?

- (One) `List Int`
- (Two) `List (Int,Int)`
- (Three) `(Int,List Int)`
- (Four) `(List Int,List Int)`
- (Five) `(Int,List (Int,Int))` ✓